

InterSystems IRIS Core Solutions Developer Specialist Exam

Practice Questions

This document contains 13 practice questions covering various topics of the exam and is designed to familiarize candidates with the style and format of the exam. As the questions presented here are not comprehensive to all exam topics, additional preparation beyond their review is recommended. The answers are provided on the last page of this document.

1) Consider the following class definition.

```
1 Class Test.Visit Extends %Persistent
2 {
3   Property NumVisits As Test.Integer;
4
5   ClassMethod Run() As %Status
6   {
7     set visit = ##class(Test.Visit).%New()
8     set visit.NumVisits = 2
9     set status = visit.%Save()
10    return status
11 }
12
13 }
```

Which of the following class definitions will return a successful status when the Run() method is executed?

- Class Test.Integer Extends %Persistent
{Parameter MINVAL = 1;}
- Class Test.Integer Extends %Persistent
{Property Integer As %Integer(MINVAL = 1);}
- Class Test.Integer Extends %Integer
{Parameter MINVAL = 1;}
- Class Test.Integer Extends %Integer
{Property Integer As %Integer(MINVAL = 1);}

2) Consider the following class definition.

```
1 Class Test.Cat Extends %Persistent
2 {
3   Property Name As %String;
4   Property color As %String *ANSWER*;
5
6   ClassMethod Run() As %Status
7   {
8     set cat = ..%New()
9     set cat.color = "calico"
10    set st = cat.%Save()
11    return st
12 }
13
14 }
```

Which TWO of the following statements, each when individually inserted at line ***ANSWER***, will return an error status when the Run() method is executed?

- (MAXLEN = 10, TRUNCATE = 0)
- (MAXLEN = 5, TRUNCATE = 1)
- (MAXLEN = 5, TRUNCATE = 0)
- (MAXLEN = 10, TRUNCATE = 1)
- (MAXLEN = 5)
- (MAXLEN = 10)

3) Consider the following two class definitions.

```
1 Class Test.MissionSuper
2 {
3
4 ClassMethod FirstMessage()
5 {
6     write !, "The mission was DARING."
7     do ..SecondMessage()
8 }
9
10
11 ClassMethod SecondMessage()
12 {
13     write !, "This is your next assignment."
14 }
15
16 }
```

****END CLASS 1***

```
1 Class Test.MissionSub extends Test.MissionSuper
2 {
3
4 ClassMethod FirstMessage()
5 {
6     write !, "The mission was a SUCCESS."
7     do ##super()
8 }
9
10 ClassMethod SecondMessage()
11 {
12     write !, "You have been promoted."
13 }
14
15 }
```

What is displayed when the command `do ##class(Test.MissionSub).FirstMessage()` is executed?

- The mission was a SUCCESS.
The mission was DARING.
This is your next assignment.
- The mission was a SUCCESS.
The mission was DARING.
You have been promoted.
- The mission was a SUCCESS.
You have been promoted.
- The mission was a SUCCESS.
This is your next assignment.
- An error message

4) Consider the following class definition.

```
1 Class Test.NewPatient Extends %Persistent
2 {
3   Property Name As %String;
4   Property Ailments As Array of %String;
5
6   ClassMethod Run()
7   {
8     set patient = ##class(Test.NewPatient).%New()
9     set patient.Name = "Seamus Jackson"
10    do patient.Ailments.SetAt("headache", "morning")
11    do patient.Ailments.SetAt("fever", "evening")
12    do patient.Ailments.SetAt("cough", "afternoon")
13    write patient.Ailments.GetNext("evening")
14 }
15
16 }
```

What is displayed when the Run() method is executed?

- fever
- cough
- 1
- headache
- 2
- 3

5) Consider the two class definitions below.

```
1 Class Test.Office Extends (%Persistent, %JSON.Adaptor)
2 {
3   Property City As %String;
4   Property Country As %String;
5 }
```

****END CLASS 1****

```
1 Class Test.Owner Extends (%Persistent, %JSON.Adaptor)
2 {
3   Property Name As %String;
4   Property Location As Test.Office;
5
6   ClassMethod Run()
7   {
8     set owner = ##class(Test.Owner).%New()
9     set TempHold = {"Name":"Artemi Zlatorov", "Location":{"City":"Moscow",
"Country":"Russia"}}
10    *ANSWER*
11    write owner.Location.City, !, owner.Location.Country
12 }
13
14 }
```

What should ***ANSWER*** be replaced with in order to have the output shown below displayed when the Run() method is executed?

Moscow
Russia

- do \$SYSTEM.OBJ.owner.Import("TempHold.json")
- do owner.%JSONImportToObject(TempHold)
- do \$SYSTEM.OBJ.owner.Load("TempHold.json")
- do owner.%JSONImport(TempHold)
- set owner.Location = TempHold
- set owner.Name = TempHold.Name
set owner.City = TempHold.City

6) Consider the following class definition.

```
1 Class Test.Date
2 {
3
4 ClassMethod Run()
5 {
6     set x = $horolog
7     set y = $piece(x, ",", 1)
8     set z = $piece(x, ",", 2)
9     *ANSWER*
10 }
11
12 }
```

Suppose the Run() method is called at 2:00 pm in the local time zone of the IRIS instance. Which TWO of the following choices, each when individually inserted in place of *ANSWER*, would display today's date?

- write \$zdate(z)
- write \$zdate(y)
- write \$zdate(x)
- write \$zdate(x,y)
- write \$zdate(x,z)

7) Consider the following class definition.

```
1 Class Test.TestIf
2 {
3
4 ClassMethod Message(a as %Integer,
5                     b as %Integer,
6                     c as %Integer,
7                     d as %Integer,
8                     e as %Integer,
9                     f as %Integer)
10 {
11     if (a >= b) && (c != d) {
12         write "Message 1"
13     }
14     elseif (e > f) {
15         write "Message 2"
16     }
17     else {
18         write "Message 3"
19     }
20 }
21
22 }
```

What will be displayed when do `##class(Test.TestIf).Message(1, 2, 3, 4, 3, 2)` is executed?

- Message 1
- Message 2
- Message 3
- An error

8) Consider the following class definition.

```
1 Class Test.Visits
2 {
3
4 ClassMethod Run(NumVis As %Integer)
5 {
6     try {
7         if (NumVis < 0){
8             set ex = ##class(%Exception.General).%New("Exception: Visits can't be
negative.", "234")
9             throw ex
10        }
11        write !, "You visited ", NumVis, " times this year."
12    }
13    catch ex {
14        if ex.%IsA("%Exception.General"){
15            write !, "Error ", ex.Code, ". " , ex.Name
16        }
17        else {
18            write !, "Something doesn't seem right."
19        }
20    }
21 }
22
23 }
```

What is displayed when the do ##class(Test.Visits).Run(-3) method is executed?

- Error 234. Exception: Visits can't be negative.
- You visited -3 times this year.
Something doesn't seem right.
- You visited -3 times this year.
Error 234. Exception: Visits can't be negative.
- You visited -3 times this year.
- Something doesn't seem right.
You visited -3 times this year.
- Error 234. Exception: Visits can't be negative.
You visited -3 times this year.

- 9) Suppose the persistent class `Test.Doctor` has a `Name` property, and has an object stored in the database with `ID = 2`.

Next suppose that two processes, A and B, are attempting to access and update the same object from the database at approximately the same time.

First, process A executes the following commands.

```
set doctor = ##class(Test.Doctor).%OpenId(2)
set doctor.Name = "Dr.Einstein"
```

Second, process B executes the following commands.

```
set doctor = ##class(Test.Doctor).%OpenId(2)
set doctor.Name = "Dr.Newton"
tstart
set st = doctor.%Save()
```

Third, process A executes the following commands.

```
set st = doctor.%Save()
write st
```

What happens, and why, when Process A executes the command `write st`?

- The ID number and "Dr. Einstein" is displayed since process A has a shared lock on the object.
- A successful status is displayed since process A has an exclusive lock on the object.
- An error status is displayed since process B has an exclusive lock on the object.
- An error status is displayed since process B has a shared lock on the object.

10) Consider the following class definition.

```
1 Class Test.GlobalAC
2 {
3
4 ClassMethod Run()
5 {
6     kill ^A
7     set ^A("a", 1, 2) = 1
8     set ^A("a", 2) = 32
9     set ^A("a") = 256
10    set ans = $data(^A("a", 1, 3))
11    write ans
12 }
13
14 }
```

What is displayed when the Run() method is executed?

- 10
- 0
- 1
- ""
- 11

11) Consider the following two class definitions and the two screenshots from the Management Portal below.

```

1 Class Test.PatientQ Extends %Persistent
2 {
3 Property Name As %String [ Required ];
4 Property City As %String [ Required ];
5 Relationship Doctor As Test.DoctorQ [ Cardinality = one, Inverse = Patients ];
6 }

```

****END CLASS 1****

```

1 Class Test.DoctorQ Extends %Persistent
2 {
3 Property Name As %String [ Required ];
4 Property City As %String [ Required ];
5 Relationship Patients As Test.PatientQ [ Cardinality = many, Inverse = Doctor ];
6 }

```

Test.PatientQ in namespace USER					Test.DoctorQ in namespace USER			
#	ID	City	Doctor	Name	#	ID	City	Name
1	1	Jackson	2	Willeke,Lola M.	1	1	Jackson	Newton,Debra Q.
2	2	Xavier	1	Goeffrion,Kristen L.	2	2	Xavier	Rotterman,Wilma N.
3	3	Pueblo	1	Xavier,Nataliya I.	3	3	Miami	Smith,Brendan M.
4	4	Xavier	3	Ingersol,Quentin B.	<i>Complete</i>			
5	5	Youngstown	2	Finn,Wilma H.				
6	6	Larchmont	2	Eisenstien,Mary G.				
7	7	Tampa	1	Jafari,Tara U.				
8	8	Bensonhurst	2	Faust,Pat Q.				
9	9	Newton	2	Ironhorse,John O.				
10	10	Zanesville	1	Young,Joe Y.				
<i>Complete</i>								

Which of the following statements will return "Xavier"?

- set doctor= ##class(Test.DoctorQ).%OpenId(3)
write doctor.Patients.GetAt(1).City
- set doctor= ##class(Test.DoctorQ).%OpenId(2, 2)
write doctor.Patients.City
- set patient= ##class(Test.PatientQ).%OpenId("3||4")
write patient.City
- set patient= ##class(Test.PatientQ).%OpenId(4)
write patient.Doctor.GetAt(1).City

12) Use the following two class definitions to answer the question below.

```
1 Class Test.Doctor Extends %Persistent
2 {
3   Property Name As %String [ Required ];
4   Property City As %String [ Required ];
5   Property State As %String [ Required ];
6 }
```

****END CLASS 1****

```
1 Class Test.DoctorQuery Extends %Persistent
2 {
3
4   ClassMethod DoctorFinder(city As %String, state As %String) As %Status
5   {
6     &sql(declare Docs cursor for
7       select Name
8       from Test.Doctor
9       where (City = :city AND State =: state)
10      order by Name)
11
12     &sql(open Docs)
13     for {
14       &sql(fetch Docs into :name)
15       *ANSWER*
16       write !, name
17     }
18     if SQLCODE = 100 {
19       write !, %ROWCOUNT, " row(s) returned."
20       set status = $$$OK
21     }
22     else {
23       set status = $$$ERROR($$$GeneralError, "SQLCODE: " _ SQLCODE _ " " _ %msg)
24     }
25     &sql(close Docs)
26     return status
27 }
28
29 }
```

Several objects from Test.Doctor exist in the database. What should ***ANSWER*** be replaced with so that the command `do ##class(Test.DoctorQuery).DoctorFinder("Boston", "MA")` displays a list of names of doctors from Boston, MA?

- if (SQLCODE '= 0) { quit }
- if (SQLCODE = 1) { quit }
- if (SQLCODE '= 100) { quit }
- if (SQLCODE '= 1) { quit }
- if (SQLCODE = 0) { quit }

13) Which of the following is a valid JDBC URL (connection string) for IRIS when using the JDBC driver?

- jdbc:IRIS://127.46.1.1:52773/
- jdbc:IRIS:setconnection:localhost:52773
- jdbc:IRIS:setconnection:localhost:51773:USER
- jdbc:IRIS://127.0.0.1:51773/USER

ANSWERS

- 1) `Class Test.Integer Extends %Integer
{Parameter MINVAL = 1;}`
- 2) `(MAXLEN = 5, TRUNCATE = 0), (MAXLEN = 5)`
- 3) The mission was a SUCCESS.
The mission was DARING.
You have been promoted.
- 4) headache
- 5) `do owner.%JSONImport(TempHold)`
- 6) `write $zdate(x),write $zdate(y)`
- 7) Message 3
- 8) Error 234. Exception: Visits can't be negative.
- 9) An error status is displayed since process B has an exclusive lock on the object.
- 10) 0
- 11) `set doctor= ##class(Test.DoctorQ).%OpenId(3)
write doctor.Patients.GetAt(1).City`
- 12) `if (SQLCODE '= 0) { quit }`
- 13) `jdbc:IRIS://127.0.0.1:51773/USER`